
 

 

How to Consume CDS View I_BillingRequestItem 

Business Case: Adding Service Contract Dates to Invoice 
Item 

Sample code: 

*   Mandatory step for all implementations: 

*   Copy the importing parameters to the changing parameters. 

    MOVE-CORRESPONDING bil_doc TO bil_doc_res. 

    MOVE-CORRESPONDING bil_doc_item TO bil_doc_item_res. 

    MOVE-CORRESPONDING bil_doc_item_contr TO bil_doc_item_contr_res. 

 

*This sample code is used to enhance customer logic for service contract related invoice/cancelled (Document Type 'F2'). 

IF ( bil_doc-sddocumentcategory = 'M'   OR              "Invoice 

     bil_doc-sddocumentcategory = 'N' ) AND             "Cancelled Invoice 

     bil_doc_item-referencesddocumentcategory = 'EBDR'. "Reference document category 

 

*   BDR and service document 

    DATA(lv_bdr)                 = bil_doc_item-referencesddocument. 

    DATA(lv_bdritem)             = bil_doc_item-referencesddocumentitem. 

    data(lv_servicecontract)     = bil_doc_item-servicedocument. 

    data(lv_servicecontractitem) = bil_doc_item-servicedocumentitem. 

 

*    Get Billing request line number 

    SELECT SINGLE referencesddocumentitem 

        FROM i_billingdocumentitembasic 

        WHERE billingdocument             = @lv_bdr 

          AND billingdocumentitem         = @lv_bdritem 

          AND referencesddocumentcategory = 'CSCT' 

         INTO @DATA(lv_record_no) . 

 

*    Get service Contract Item Start/End date 

    SELECT SINGLE servicecontritemstartdatetime, 

                  servicecontritemenddatetime 

        FROM i_servicedocumentitem 

       WHERE servicedocument     = @lv_servicecontract 

         AND servicedocumentitem = @lv_servicecontractitem 

        INTO ( @DATA(lv_contrstart), 

               @DATA(lv_contraend) ). 

 

*    Get settlement Start/End date 

    SELECT SINGLE settlementstartdatetime, 

                  settlementenddatetime 

        FROM i_billingrequestitem 

       WHERE servicedocument     = @lv_servicecontract 



         AND servicedocumentitem = @lv_servicecontractitem 

         AND billingrequestitem  = @lv_record_no 

        INTO ( @DATA(lv_settlementstart), 

               @DATA(lv_settlementend) ). 

 

    CONVERT TIME STAMP lv_contrstart      TIME ZONE 'UTC' INTO DATE bil_doc_item_res-yy1_demosrvcstart_bdi. 

    CONVERT TIME STAMP lv_contraend       TIME ZONE 'UTC' INTO DATE bil_doc_item_res-yy1_demosrvcend_bdi. 

    CONVERT TIME STAMP lv_settlementstart TIME ZONE 'UTC' INTO DATE bil_doc_item_res-yy1_demosettlfrom_bdi. 

    CONVERT TIME STAMP lv_settlementend   TIME ZONE 'UTC' INTO DATE bil_doc_item_res-yy1_demosettleto_bdi. 

ENDIF. 

 

 


